3. An Overview of Graphics

One can use Mathematica to make 2 D and 3 D graphics. It is perhaps the most straightforward and, at the same time, the most complicated section. It is straightforward since all the commands and properties one can find in the Help Browser. However, the complicated part is to find the property one needs among hundreds of similar ones. Below there are only examples of some of the most commongly used graphic features. Moreover, the older versions of *Mathematica* might have different names and commands. The following pictures are drawn in *Mathematica* 7. The pictures are mostly self-explanatory and for the useage of unknown functions the reader is referred to the documentation center.

```
Plot[Sin[x^2], {x, 0, 4}, PlotRange → {0, 1.2}]
```



```
ParametricPlot3D[{Sin[z] Sin[t], Sin[z] Cos[t], z},
 {z, -π, π}, {t, 0, 2 π}, PlotRange → {{0, 1}, {0, 1}, {0, 1}}]
```




```
Plot[x^5 - 4.5 x^4 + 2.1 x^2 - 7, {x, -10, 14}, PlotRange → All]
```



```
Graphics[Circle[{0, 0}, 1], PlotLabel → "circle"]
```




```
Graphics[Circle[{0, 0}, 1], AspectRatio → 1 / 2, PlotLabel → "ellipse"]
```



```
Graphics[Circle[{0, 0}, 1], Axes → Automatic]
```




```
Plot[Sin[x], {x, 0, 2 π}, Ticks → {{0, π/2, π, 3 π/2, 2 π}, Automatic}]
```




```
Plot[Sin[x], {x, 0, 2 π}, AxesStyle → {RGBColor[1, 0, 0], Thickness[0.01]}]
```



```
Plot[Sin[x], {x, 0, 2 π}, Frame → True]
```




```
Plot[Log[x] + Sin[x + √2 Sin[x]], {x, 0, 10}, GridLines → Automatic]
```




```
data = Table[Sin[x] + Random[Real, {-0.1, 0.1}], {x, 0, 2 π, 0.1}]
```

```
{-0.0994549, 0.0025176, 0.169147, 0.35959, 0.426721, 0.504849, 0.5569, 0.629977,
 0.689323, 0.700252, 0.881904, 0.895217, 1.0026, 1.00545, 1.06306, 1.009,
 0.983381, 1.07258, 0.893839, 0.900578, 0.940057, 0.831201, 0.780229,
 0.703249, 0.705678, 0.563779, 0.416756, 0.420854, 0.4279, 0.279133,
 0.150118, -0.0507052, -0.0374291, -0.134788, -0.186977, -0.347078,
 -0.392135, -0.448773, -0.520907, -0.595565, -0.790225, -0.918132,
 -0.800616, -0.878242, -0.915784, -0.945376, -0.994464, -1.01954, -1.09056,
 -1.01561, -0.960952, -0.838909, -0.970764, -0.805304, -0.68379, -0.626348,
 -0.639521, -0.64668, -0.544192, -0.398389, -0.238055, -0.25922, -0.15363}
```

```
ListPlot[data]
```




```
ListLinePlot[data]
```




```
Plot[{BesselJ[2, z], BesselJ[3, z], BesselJ[4, z]},
 {z, -10, 16}, PlotStyle → {Hue[0.5], Hue[0.6], Hue[0.7]}]
```



```
Plot3D[Sin[x y], {x, 0, 2 π}, {y, 0, 2 π}, PlotPoints → 40]
```




```
Plot3D[Sin[x y], {x, 0, 2 π}, {y, 0, 2 π}, PlotPoints → 10]
```



```
Plot3D[4 e^-(x-1)^2-(y-2)^2, {x, -2, 2}, {y, -3, 3},
 PlotRange -> All, AxesLabel -> {"x", "y", "z"}]
```




```
Show[%, ViewPoint -> {1.2, 1.2, 1.2}]
```



```
Plot3D[4 e^-(x-1)^2-(y-2)^2, {x, -2, 2}, {y, -3, 3},
 PlotRange -> All, AxesLabel -> {"x", "y", "z"}]
```




```
Show[%, ViewVertical → {1, 0, 0}]
```



```
Plot[Sin[x], {x, 0, 2 π}, Background → GrayLevel[0.5]]
```




```
Quit[]
```

```
PolyhedronData["Dodecahedron"]
```



```
ListPlot3D[Table[Mod[y, x], {x, 10, 30}, {y, 10, 30}]]
```



The function Show is used to combine several graphics togather.

```
plot1 = Plot[Sin[x], {x, -2 Pi, 2 π}, PlotStyle → Red];
```

```
plot2 = Plot[Sin[2 x], {x, -2 Pi, 2 π}, PlotStyle → Green];
```

```
plot3 = Graphics[{Yellow, Circle[{0, 0}, 1]}];
```

```
Show[plot1, plot2, plot3, AspectRatio → Automatic]
```



Graphics[primitives, options]represents a two-dimensional graphical image (circle, disc, point, line, polygon, ...).

```
vertices = {{0, -1}, {1, 0}, {0, 1}, {-1, 0}, {0, -1}};
```

```
p = Graphics[{RGBColor[1, 0, 0], Polygon[vertices]}];
```

```
l = Graphics[{Thickness[.02], RGBColor[0, 0, 1], Line[vertices]}];
```

```
Show[p, l]
```



```
p1 = Plot[Sin[x], {x, 0, 2 π}]
```




```
p2 = Plot[Sin[2 x], {x, 0, 2 π}];
```

```
GraphicsGrid[{{p1, p2}}]
```



```
GraphicsGrid[{{p1}, {p2}}]
```




```
RandomReal[{0, 1}, {24, 2}]
```

```
{{0.455485, 0.477362}, {0.70587, 0.75487}, {0.554577, 0.621808},
 {0.801264, 0.342213}, {0.348274, 0.598254}, {0.378546, 0.520606},
 {0.598194, 0.411677}, {0.856736, 0.864341}, {0.807134, 0.0230063},
 {0.878315, 0.454539}, {0.874411, 0.0559939}, {0.32683, 0.456437},
 {0.781892, 0.647471}, {0.975297, 0.293457}, {0.453405, 0.246344},
 {0.452779, 0.784384}, {0.500323, 0.796357}, {0.218455, 0.890308},
 {0.128262, 0.63954}, {0.927231, 0.22219}, {0.287996, 0.846428},
 {0.823864, 0.134597}, {0.773817, 0.869142}, {0.264906, 0.357928}}
```

```
Graphics[Line[RandomReal[{0, 1}, {24, 2}]]]
```



```
Graphics[
 {Hue[.77], Rectangle[{0, 0}, {5, 1}], Hue[.17], Rectangle[{3, 0}, {4, 2}]}]
```



```
Graphics3D[
 {Cuboid[{0, 0, 0}], Cuboid[{0, 0, 1}], Cuboid[{0, 1, 1}], Cuboid[{1, 1, 1}]}]
```



```
Graphics[{{Circle[{0, 0}, 1], Circle[{0, 0}, {2, 1}],
   Circle[{3/2, -1/4}, Sqrt[5]/4, {1/2, 2}]}, AspectRatio -> Automatic, Axes -> Automatic]
```



```
InscribedCircleData[pA : {_, _}, pB : {_, _}, pC : {_, _}] :=
 Module[{AB, BC, AC, a, b, c, s, pP, pQ, AP, BQ,
   p, q, ps, qs, pqs, incenter, inradius}, AB = pB - pA;
  BC = pC - pB;
  AC = pC - pA;
  a = √BC.BC ;
  b = √AC.AC ;
  c = √AB.AB ;
  AP = pB + p BC - pA;
  BQ = pA + q AC - pB;
  ps = Solve[AP.AB / c == AP.AC / b, p][[1, 1]];
  qs = Solve[BQ.BC / a == BQ.(-AB) / c, q][[1, 1]];
  pP = pB + p BC /.ps;
  pQ = pA + q AC /.qs;
  pqs = Solve[pA + p (pP - pA) == pB + q (pQ - pB), {p, q}][[1]];
  incenter = pA + p (pP - pA) /.pqs;
  s = 1/2 (a + b + c);
  inradius = √((s - a) (s - b) (s - c) / s) ; {incenter, inradius}]

InscribedCircle[pA : {_, _}, pB : {_, _}, pC : {_, _}] := Graphics[
  {Line[{pA, pB, pC, pA}], Circle[Sequence @@ InscribedCircleData[pA, pB, pC]]},
  AspectRatio -> Automatic, PlotRange -> All, Frame -> True]
```

```
InscribedCircle[{1.8, 6.8}, {3.1, 1.1}, {6.4, 2.4}]
```




```
Graphics[
 {{RGBColor[0, 0, 1], Disk[{0, 0}, {2, 1}]}, {RGBColor[0, 1, 1], Disk[{0, 0}, 1]},
  {RGBColor[1, 1, 0], Disk[{0, 0}, 2, {0, 1}]}},
 AspectRatio → Automatic, Axes → Automatic]
```



```
Graphics[{Text["Left", {-1, 0}, {-1, 0}], Text["Right", {1, 0}, {1, 0}],
  Text["Above", {0, 1}, {0, -1}], Text["Below", {0, -1}, {0, 1}],
  {PointSize[.075], RGBColor[1, 0, 0], Point[{0, 0}]}}, PlotRange → All]
```




```
ListPlot[Table[{x, EulerPhi[x]}, {x, 15}], PlotStyle → PointSize[0.02]]
```



```
Graphics[{Dashing[{0.15, 0.05}], Circle[{0, 0}, 1]}, AspectRatio → Automatic]
```



```
Play[Sin[2^t], {t, 11, 14}]
```



```
Animate[Plot[Sin[n x], {x, 0, 2 Pi}, Axes -> False], {n, 1, 3, 1}]
```




```
g = ParametricPlot3D[
 {x, Cos[t] Sin[x] , Sin[t] Sin[x]},
      {x, -Pi, Pi}, {t, 0, 2Pi},
      Axes -> False, Boxed -> False]
```



```
Quit[]
```

```
Graphics[{Arrow[{{0, 0},{1, 1}}],
    Hue[0], Arrow[{{.75, .25},{.25, .75}}]}]
```



```
Plot[Sin[x], {x, 0, 2Pi},
    Epilog -> {Arrow[{{4, .25}, {Pi/2, 1}}],
     Text["Here", {4, .15}, {0, -1}]}
]
```



```
ContourPlot[x^2 + 2 y^2 == 3, {x, -2, 2},{y,-2,2}]
```




```
ContourPlot[{(x^2 + y^2)^2 == (x^2 - y^2),
  (x^2 + y^2)^2 == 2 x y}, {x,-2,2},{y,-2,2}
]
```



```
RegionPlot[ 1 ≤ (x + 2 y)^2 + 4 y^2 ≤ 4, {x, -3, 3}, {y, -3, 3} ]
```




```
RegionPlot3D[ x^2 + y^2 + z^2 ≤ 1 ⋀ 3 x^2 + 3 y^2 ≤ z^2, {x, -1, 1}, {y, -1, 1}, {z, -1, 1}]
```



```
VectorPlot[{Sin[x], Cos[y]}, {x, 0, Pi}, {y, 0, Pi}]
```




```
RevolutionPlot3D[
        Sin[x], {x, 0, 2 Pi}]
```



```
RevolutionPlot3D[{1.1 Sin[u], u^2},
  {u, 0, 3 Pi/2}, BoxRatios -> {1, 1, 2}]
```




```
RevolutionPlot3D[x^2, {x, 0, 1},
    RevolutionAxis -> {1, 1, 1}]
```



```
FinancialData["GE", "Price"]
```

```
20.595
```

```
DateListLogPlot[FinancialData["^DJI", All], Joined → True, Filling → Bottom]
```




```
Plot[Table[BesselJ[n, x], {n, 4}], {x, 0, 15}, Filling → Axis, Evaluated -> True]
```



```
ContourPlot[Sin[3 x] Cos[x + y], {x, 0, 3}, {y, 0, 3},
 ContourLabels → Automatic, ColorFunction → "Pastel"]
```




```
ParametricPlot3D[
 {Cos[v] + 0.3 Sin[3 u] + 0.04 Sin[20 v], u, Sin[v] + 0.3 Cos[3 u] + 0.04 Sin[20 v]},
 {u, -π, π}, {v, -π, π}, PlotPoints → 100,
 PlotStyle → {Orange, Specularity[White, 10]}, Axes → None, Mesh → None]
```



One can innsert a plot into a disk:

```
Graphics[{LightGray, Disk[], Inset[Plot[Tan[x], {x, -3, 3}]]}]
```



One can insert an expression in a graphic:

```
Graphics[{Circle[], Inset[X^2 + Y^2 == 1, {0, 0}]}]
```




```
solution2 = NDSolve[{x''[t] + x[t]^3 == Sin[t], x[0] == x'[0] == 0}, x, {t, 0, 50}]
```

```
{{x → InterpolatingFunction[{{0., 50.}}, <>]}}
```

```
ParametricPlot[{x[t], x'[t]} /. solution2, {t, 0, 50}]
```



```
Block[{f = Cos[x + I y]},
 ParametricPlot[Evaluate[{Re[f], Im[f]}], {x, -Pi, Pi}, {y, -2, 2}]]
```